NATIONAL RADIO ASTRONOMY OBSERVATORY
Charlottesville, Virginia

Electronics Division Internal Report No. 131

COMPUTATION OF APERTURE EFFICIENCY AND RADIATION PATTERN FOR A CASSEGRAIN ANTENNA

T. W. Leonard and P. J. Napier

AUGUST 1973

NUMBER OF COPIES: 150

COMPUTATION OF APERTURE EFFICIENCY AND RADIATION PATTERN FOR A CASSEGRAIN ANTENNA

1. Introduction

This report describes a program which calculates taper, spillover, and phase efficiencies, a relative aperture voltage distribution, and the radiation pattern of a cassegrain antenna system given the feed power pattern data and the cassegrain geometry. The program language is Basic and is written for use on the Hewlett-Packard 9830A calculator.

Circular symmetry is assumed throughout the program. Plots of the aperture voltage distribution and the radiation pattern are provided in addition to numerical tables of each. A provision is included for accounting for aperture blockage due to the subreflector or prime-focus structure. This program can be used for a prime focus antenna if the angle subtended by the subreflector is chosen to be the same as the angle subtended by the main dish. In this case the cassegrain magnification factor is unity and all calculations are identical to those for a prime focus system. A brief description of the theory for the major parts of the program is given. The program is then described and examples of input and output data are provided.

A future memo will describe how the program may be applied to a cassegrain system in which the reflector surfaces have been shaped for high aperture efficiency.

2. Taper Efficiency

The taper efficiency of a parabolic reflector system is defined as the ratio of the effective area of the paraboloid with tapered illumination to the physically projected areas of the paraboloid, and is shown by Weinreb[1] to be expressable as

$$\eta_T = 32 \left(\frac{F}{D}\right)^2 \frac{\left|\int_0^{\theta_0} \sqrt{G(\theta)}\ \tan \theta/2\ d\theta\right|^2}{\int_0^{\theta_0} G(\theta) \sin \theta\ d\theta} \qquad (1)$$

for a prime focus dish, with $\theta_0 = 2 \tan^{-1} \frac{D}{4F}$ edge angle, F/D = the focal length to diameter ratio, and $G(\theta)$ is the power pattern of the spherical wave propagating towards the parabola from its focus. Figure 1 defines the important parameters of a cassegrain system.

Potter[2] derives two equations which allow equation (1) to be applied to a cassegrain system. They are the relationship of θ and γ (see Figure 1)

$$\tan \left(\frac{\theta}{2}\right) = M \tan \left(\frac{\gamma}{2}\right) \tag{2}$$

and the power pattern weighting factor

$$G(\theta) = \left(\frac{\sin \gamma}{\sin \theta}\right)^2 F(\gamma) \tag{3}$$

where $M = \frac{L_R}{L_V}$ is the magnification of the cassegrain system[3], and $F(\gamma)$ is the feed radiation pattern.

Using (2) and (3) we obtain

$$\theta = 2 \tan^{-1} \left[M \tan \left(\frac{\gamma}{2}\right)\right] \tag{4}$$

and

$$G(\theta) = \frac{\sin^2(\gamma) \; [F(\gamma)]}{\sin^2[2 \tan^{-1}(M \tan (\frac{\gamma}{2}))]} \tag{5}$$

Substituting equations (4) and (5) into equation (1), and using the logarithmic expansion for the arc tan[4], it can be shown that

$$\eta_T = 32 \left(\frac{MF}{D}\right)^2 \frac{\left| \int_0^{\gamma_0} \sqrt{F(\gamma)} \; \tan \frac{\gamma}{2} \, d\gamma \right|^2}{\int_0^{\gamma_0} F(\gamma) \sin \gamma \, d\gamma} \tag{6}$$

Thus, for a cassegrain system, the taper efficiency is the same as if the feed had been used to illuminate a parabola of focal length MF. The physical reason for this is that the subreflector tends to "bundle" energy towards the outside of the parabola (as shown by equation (3)) so that the effect of space-taper on the feed pattern is offset and a higher taper efficiency results

3. Spillover Efficiency

The spillover calculation is identical to that of a prime focus dish. The assumption is made that all of the radiation illuminating the subreflector dish also falls upon the main reflector after striking the subreflector. The spillover efficiency, which is the ratio of energy incident upon the subreflector to the total energy emitted by the feed, can be expressed as[1]

$$\eta_s = \frac{\int_0^{\gamma_0} F(\gamma) \sin \gamma \, d\gamma}{\int_0^{\pi} F(\gamma) \sin \gamma \, d\gamma} \tag{7}$$

where $\gamma_0 \equiv$ the angle subtended by the subreflector.

The program sets all points of the feed pattern which are not assigned a value to 60 dB below the feed axis reference point.

4. Phase Efficiency

Phase efficiency can be defined as the ratio of the on axis power with phase errors to the on axis power with no phase errors. It can be expressed as

$$\eta_p = \frac{\left| \int_0^{\gamma_0} \sqrt{F(\gamma)} \, \varepsilon^{j\phi} \tan \gamma/2 \, d\gamma \right|^2}{\left| \int_0^{\gamma_0} \sqrt{F(\gamma)} \tan \gamma/2 \, d\gamma \right|^2} \tag{8}$$

Small linear phase shifts across the aperture tend to merely shift the true direction of the beam rather than degrade the efficiency of the antenna. Therefore, phase errors should be taken as the deviation in phase from the best fit linear phase shift across the aperture. If the feed phase information is not available, this section of the program may be by-passed.

5. Aperture Distribution Calculation

If the illuminating pattern is known for a prime focus dish, the corresponding aperture voltage distribution can be calculated as a function of the reflector radius R[1]

$$E(R) = \sqrt{G(\theta)} \cos^2 (\theta/2) \tag{9}$$

R is calculated from the definition of a parabolic surface[5] (see figure 1)

$$\rho = \frac{2F}{1 + \cos \theta} \tag{10}$$

Thus: $$R = \frac{2F \sin(\theta)}{1 + \cos\theta} \qquad (11)$$

The calculation for a cassegrain system is performed by substituting equations (2), (3) and (11) into equation (9).

The aperture distribution is calculated for only those points in the aperture which correspond to feed pattern points falling on the subreflector. This number of points, which depends on the feed pattern increment of angle and the angular extent of the subreflector, is usually an insufficient number of points for accurate evaluation of an integral in the radiation pattern section of the program. A polynomial regression routine is used to best fit a 6th degree polynomial to the aperture data using a least-squares fit. This section is taken from the Hewlett Packard Math Pack tape with some modifications. The polynomial is then used to make 51 estimates of aperture voltage at equal increments across the aperture. At least seven feed pattern points must fall on the subreflector for a 6th degree polynomial fit.

6. Radiation Pattern

The relative voltage radiation pattern of an antenna with a circularly symmetric aperture can be expressed as[5]

$$F(u) = \int_0^1 f(R)\, J_o(uR)\, R\, dR \qquad (12)$$

where $u = \frac{\pi D \sin\theta}{\lambda}$, f(R)=aperture voltage distribution and R=normalized radians.

This integral is evaluated for u = 0 to 10 in increments of 0.5 u. The resulting normalized points of the radiation pattern are printed in tabular form and are also plotted. The Bessel function evaluation is taken from the Hewlett Packard Math Pack program with modifications.

7. Program Description

The program is located on tape files 1-8, beginning with file 1. The program's input format is conversational. It will ask for all data required at the beginning of the program. The feed pattern information should be expressed in DB below a zero DB reference. A maximum of 151 feed pattern data points may be used. The increment of angle should be small enough so that several points fall on the subreflector for accurate calculations. Phase points and the angle increment are expressed in degrees. For phase calculations, the minimum number of phase points used should equal the total number of feed points falling on the subreflector plus one. A maximum of 50 phase points may be used and they must be incremented identically to the feed pattern points.

For prime focus dish calculations, the angle subtended by the subreflector should be set equal to the value of 0 given by

$$\theta = 2 \tan^{-1} D/4F \tag{13}$$

This value insures that the magnification factor M=1 so the taper efficiency integral will be correct.

The printed output lists the taper, spillover, and phase efficiencies and their product. It then lists values of aperture voltage and radius, both normalized to maximum of unity. Coefficients of the 6th degree polynomial are then printed, along with a number labeled, R Squared" which, if it exceeds unity, indicates an inaccurate curve fit. Next, 51 estimates of the aperture distribution are listed while they are simultaneously plotted. Values of normalized radiation pattern voltage are then listed and plotted.

8. Efficiency Program Verification

To determine the accuracy of the integration techniques and the overall efficiency sections, a feed pattern resulting in known efficiencies for a standard Cassegrain system was chosen.

The feed pattern was:

$$F(\gamma) = \varepsilon^{2a\gamma} \cos^2 \frac{\gamma}{2} \tag{14}$$

and the angle subtended by the subreflector was chosen as γ_o=15 degrees. Feed data points were chosen at increments of one degree, and data points were fed in out to 60 degrees. A comparison of the program calculated values to the known values follows:

	PROGRAM%	KNOWN%
Taper Efficiency	98.3572%	98.3570%
Spillover Efficiency	30.52%	30.45%
Total Efficiency	30.02%	29.95%

9. Example Program

A sample program follows to demonstrate typical input and output data. The example used is for the 140-ft. diameter dish with the following Cassegrain geometry.

F/D = 0.4286 γ_0 = 7.14 degrees subreflector diameter = 10.4 ft.

Data points were taken from graphs of the feed power pattern and the feed phase pattern. Seventy-five feed pattern points and sixteen phase points were taken. Percent radius blockage was 7.4% (this is the ratio of subreflector diameter to main reflector diameter). Feed patterns were sampled at increments of 0.5 degrees. A listing of the input data appears in Figure 2. Output data is listed in Figure 3.

The program's plot of the aperture distribution appears in Figure 4. The X axis is normalized radius and the Y axis is normalized voltage. Note the effect of the subreflector blockage for values of R less than 0.08.

The radiation pattern plot appears in Figure 5. The X axis is u=0 to 10 and the Y axis is the normalized pattern voltage. The X axis can be converted to angle θ at a particular wavelength λ using the definition

$$u = \frac{\pi D \sin \theta}{\lambda}$$

REFERENCES

[1]S. Weinreb and S. Jansson, "Antenna Feed Efficiency and Spillover Calculation Program", NRAO Electronics Division Internal Report No. 93; June 1970.

[2]P. D. Potter, "Aperture Illumination and Gain of a Cassegrain System", IEEE Transactions on Antennas and Propagation (Communication), Vol. AP-11, pp.373-375; May 1963.

[3]P. W. Hannan, "Microwave Antennas Derived from the Cassegrain Telescope", IRE Transactions on Antennas and Propagation, Vol. AP-9, pp.140-153; March 1961.

[4]M. Abramowitz and I. A. Stegun, Handbook of Mathematical Functions, Dover Publications, New York, pp.81; 1965.

[5]S. Silver, Microwave Antenna Theory and Design, M.I.T. Radiation Laboratory Series, McGraw-Hill Book Co., New York, Vol. 12, pp.192-195 & 415-417; 1949



FIGURE 1 - CASSEGRAIN GEOMETRY

```
RUN
FOR PHASE CALCULATION, INPUT 1
IF PHASE NOT DESIRED, INPUT 0
?1
F/D RATIO?.4286
PERCENT RADIUS BLOCKAGE?7.4
ANGLE SUBTENDED BY SUBREFLECTOR?7.14
INCREMENT OF ANGLE?0.5
NUMBER OF FEED DATA POINTS?75
INPUT DATA POINTS? 0
 0.1
 0.2
 0.4
 0.7
 0.9
 1.4
 1.9
 2.4
 3.2
 4.2
 4.7
 5.7
 6.5
 7.4
 8.4
 9.7
 10.8
 12.3
 13.6
 15.2
 16.7
 18.1
 20.7
 23.7
 27
 32
 36.2
 36.2
 36.2
 34.6
 33.2
 31.4
 30.6
 29.8
 29.3
 29.4
 29.5
 29.9
 30.5
 31.2
 32.4
 33.2
 33.7
 34.6
 34.8
 34.7
 34.6
 34.2
 34.2
 34.5
 35
 36.2
 36.2
 36.2
 36.2
 36.2
 36.2
 36.2
 36.2
 36.2
 36.2
 36.2
 36.2
 35.5
 35
 34.2
 33.8
 33.5
 33.4
 34.3
 34.7
 35.2
 35.7
 36.2
NUMBER OF PHASE POINTS?16
INPUT PHASE IN DEGREES?
 0
 0
 0
 0
 0
 0.5
 1
 1.5
 2
 3
 4.5
 6.5
 8
 11
 14
 18
SET UP PLOTTER
```

FIGURE 2 - INPUT DATA

APERTURE EFFICIENCY

F/D RATIO=	0.4286	MAGNIFICATION=	9.3493
TAPER EFFICIENCY=	93.55%	SPILLOVER EFFICIENCY=	81.66%
PHASE EFFICIENCY=	99.45%	TOTAL EFFICIENCY=	75.97%

APERTURE VOLATGE DISTRIBUTION

RADIUS R	VOLTAGE E(R)	PHASE P
R= 0.0000	E(R)= 1.0000	P= 0.0
R= 0.0699	E(R)= 0.9885	P= 0.0
R= 0.1399	E(R)= 0.9772	P= 0.0
R= 0.2098	E(R)= 0.9548	P= 0.0
R= 0.2798	E(R)= 0.9223	P= 0.0
R= 0.3497	E(R)= 0.9011	P= 0.5
R= 0.4197	E(R)= 0.8506	P= 1.0
R= 0.4897	E(R)= 0.8028	P= 1.5
R= 0.5597	E(R)= 0.7577	P= 2.0
R= 0.6298	E(R)= 0.6908	P= 3.0
R= 0.6998	E(R)= 0.6154	P= 4.5
R= 0.7699	E(R)= 0.5808	P= 6.5
R= 0.8400	E(R)= 0.5174	P= 8.0
R= 0.9102	E(R)= 0.4716	P= 11.0
R= 0.9803	E(R)= 0.4250	P= 14.0

COEFFICIENTS

B(0)=	0.9994
B(1)=	-0.0327
B(2)=	-1.6362
B(3)=	6.0954
B(4)=	-15.1281
B(5)=	15.9707
B(6)=	-5.8596

R SQUARE = 0.999218447

R= 0	E(R)= 0
R= 0.02	E(R)= 0
R= 0.04	E(R)= 0
R= 0.06	E(R)= 0
R= 0.08	E(R)= 0.988888696
R= 0.1	E(R)= 0.984528906
R= 0.12	E(R)= 0.979715088
R= 0.14	E(R)= 0.974506167
R= 0.16	E(R)= 0.968934871
R= 0.18	E(R)= 0.963012102
R= 0.2	E(R)= 0.956731049
R= 0.22	E(R)= 0.950071024
R= 0.24	E(R)= 0.943001027
R= 0.26	E(R)= 0.935483049
R= 0.28	E(R)= 0.927475095
R= 0.3	E(R)= 0.918933942
R= 0.32	E(R)= 0.909817632
R= 0.34	E(R)= 0.900087681
R= 0.36	E(R)= 0.889711033
R= 0.38	E(R)= 0.878661736
R= 0.4	E(R)= 0.866922350
R= 0.42	E(R)= 0.854485082
R= 0.44	E(R)= 0.841352656
R= 0.46	E(R)= 0.827538910
R= 0.48	E(R)= 0.813069126
R= 0.5	E(R)= 0.797980080
R= 0.52	E(R)= 0.78231984
R= 0.54	E(R)= 0.766147274
R= 0.56	E(R)= 0.749531304
R= 0.58	E(R)= 0.732549880
R= 0.6	E(R)= 0.715288688
R= 0.62	E(R)= 0.697839590
R= 0.64	E(R)= 0.680298789
R= 0.66	E(R)= 0.662764724
R= 0.68	E(R)= 0.645335705
R= 0.7	E(R)= 0.628107262
R= 0.72	E(R)= 0.611169239
R= 0.74	E(R)= 0.594602606
R= 0.76	E(R)= 0.578476011
R= 0.78	E(R)= 0.562842056
R= 0.8	E(R)= 0.547733301
R= 0.82	E(R)= 0.533158006
R= 0.84	E(R)= 0.519095598
R= 0.86	E(R)= 0.505491863
R= 0.88	E(R)= 0.492253882
R= 0.9	E(R)= 0.479244678
R= 0.92	E(R)= 0.466277614
R= 0.94	E(R)= 0.453110503
R= 0.96	E(R)= 0.439439456
R= 0.98	E(R)= 0.424892463
R= 1	E(R)= 0.409022697

FIGURE 3 - OUTPUT DATA



FIGURE 4 - APERTURE DISTRIBUTION


FIGURE 5 - RADIATION PATTERN
F(U)
U
1.0
0.9
0.8
0.7
0.6
0.5
0.4
0.3
0.2
0.1
0.0
0
1
2
3
4
5
6
7
8
9
10

PROGRAM LISTING

TAPE FILE NO. 1

```
10 RAD
20 DIM GS[150],HS[50],ES[51],RS[51],C[36],B[8]
30 FOR I=1 TO 150
40 G[I]=60
50 NEXT I
60 FOR I=1 TO 50
70 H[I]=0
80 NEXT I
90 DISP "FOR PHASE CALCULATION, INPUT 1"
100 WAIT 3000
110 DISP "IF PHASE NOT DESIRED, INPUT 0"
120 WAIT 3000
130 INPUT X5
140 DISP "F/D RATIO";
150 INPUT F
160 DISP "PERCENT RADIUS BLOCKAGE";
170 INPUT H9
180 DISP "ANGLE SUBTENDED BY SUBREFLECTOR";
190 INPUT E
200 DISP "INCREMENT OF ANGLE";
210 INPUT A1
220 DISP "NUMBER OF FEED DATA POINTS";
230 INPUT D1
240 DISP "INPUT DATA POINTS";
250 FOR I=1 TO D1
260 INPUT G[I]
270 PRINT G[I]
280 NEXT I
290 IF X5=1 THEN 310
300 GOTO 390
310 DISP "NUMBER OF PHASE POINTS";
320 INPUT D3
330 DISP "INPUT PHASE IN DEGREES";
340 FOR I=1 TO D3
350 INPUT H[I]
360 PRINT H[I]
370 H[I]=H[I]*PI/180
380 NEXT I
390 DISP "SET UP PLOTTER"
400 WAIT 4000
410 FOR I=1 TO 150
420 G[I]=1/(10↑(G[I]/10))
430 NEXT I
440 M=1/(4*F*TAN((E*PI/180)/2))
450 N=(E/A1)+1.5
460 A3=A1*PI/180
470 E1=E*PI/180
480 M1=INT((E1/A3)+0.5)
490 P=(E1-M1*A3)/A3
500 LINK 2
```

TAPE FILE NO. 2

```
10 RAD
20 DIM GS[150],HS[50],ES[51],RS[51],C[36],B[8]
30 R3=V1=O1=U1=S1=T1=0
40 FOR I=2 TO N
50 Q=A3*(I-1)
60 V=SQR(G[I])*TAN(Q/2)*A3
70 V1=V1+V
80 U=G[I]*SIN(Q)*A3
90 U1=U1+U
100 NEXT I
110 U1=U1-U+U*(0.5+P)
120 V1=V1-V+V*(0.5+P)
130 N1=((32*(F*M)↑2*V1↑2)/U1)*100
140 I=1
150 I=I+1
160 Q=A3*(I-1)
170 O=G[I]*SIN(Q)*A3
180 O1=O1+O
190 IF I=150 THEN 220
200 IF Q>PI THEN 270
210 GOTO 150
220 Q=Q+A3
230 IF Q>PI THEN 270
240 R1=SIN(Q)*A3*1E-06
250 R3=R3+R1
260 GOTO 220
270 R3=R3+O1
280 N2=(U1/R3)*100
290 PRINT "                                    APERTURE EFFICIENCY"
300 WRITE (15,310)F,M
310 FORMAT 2/,3X,"F/D RATIO=",8X,F7.4,19X,"MAGNIFICATION=",7X,F8.4
320 WRITE (15,330)N1,N2
330 FORMAT 3X,"TAPER EFFICIENCY=",F7.2,"%",19X,"SPILLOVER EFFICIENCY=",F7.2,"%"
340 IF X5=1 THEN 390
350 N3=N1*N2/1E+02
360 WRITE (15,370)N3
370 FORMAT /,25X,"TOTAL EFFICIENCY=",F7.2,"%"
380 GOTO 540
390 FOR I=2 TO N
400 Q=A3*(I-1)
410 B9=SQR(G[I])*TAN(Q/2)*A3
420 T=B9*COS(H[I])
430 T1=T1+T
440 S=B9*SIN(H[I])
```

```
450 S1=S1+S
460 NEXT I
470 T1=T1-T+T*(0.5+P)
480 S1=S1-S+S*(0.5+P)
490 T9=T1↑2+S1↑2
500 T8=(T9/V1↑2)*100
510 N3=N1*N2*T8/1E+04
520 WRITE (15,530)T8,N3
530 FORMAT 3X,"PHASE EFFICIENCY=",F7.2,"%',19X,"TOTAL EFFICIENCY=",4X,F7.2,"%"
540 LINK 3
```

TAPE FILE NO. 3

```
10 RAD
20 DIM G$[150],H$[50],E$[51],R$[51],C[36],B[8]
30 E[1]=1
40 R[1]=0
50 E1=(1/M)*SQR(G[1])
60 WRITE (15,70)
70 FORMAT 3/,25X,"APERTURE VOLATGE DISTRIBUTION"
80 PRINT
90 PRINT
100 PRINT "      RADIUS  R                    VOLTAGE E(R)"TAB64"PHASE P"
110 PRINT
120 PRINT TAB5"R= 0.0000"TAB33"E(R)= 1.0000"TAB64"P=  0.0"
130 FOR I=2 TO N
140 Q=A3*(I-1)
150 B=2*ATN(M*TAN(Q/2))
160 E[I]=((SIN(Q)/SIN(B))*SQR(G[I])*(COS(B/2))↑2)/E1
170 R[I]=4*F*SIN(B)/(1+COS(B))
180 NEXT I
185 FOR I=1 TO 50
186 H[I]=H[I]*(180/PI)
187 NEXT I
190 FOR I=2 TO N
200 WRITE (15,210)R[I],E[I],H[I]
210 FORMAT 5X,"R=",F7.4,19X,"E(R)=",F7.4,19X,"P=",F5.1
220 NEXT I
230 LINK 4
```

TAPE FILE NO. 4

```
10 DIM C[36],B[8],E$[51],R$[51]
20 M8=N
30 FOR I=1 TO 8
40 C[I]=B[I]=0
50 NEXT I
60 FOR I=9 TO 36
70 C[I]=0
80 NEXT I
90 B[1]=1
100 W=N=S1=S2=S3=S4=S5=0
110 D2=6
120 FOR K=1 TO M8
130 B[2]=R[K]
140 Y=E[K]
150 Y=FNX1
160 NEXT K
170 REDIM R[1]
180 LINK 5
190 DEF FNX(Z)
200 FOR I=2 TO D2
210 B[I+1]=B[I]*B[2]
220 NEXT I
230 B[D2+2]=Y
240 R=0
250 FOR I=1 TO D2+2
260 FOR J=I TO D2+2
270 R=R+1
280 C[R]=C[R]+B[I]*B[J]*Z
290 NEXT J
300 NEXT I
310 S1=S1+B[2]*Z
320 S2=S2+B[2]↑2*Z
330 S3=S3+Y*Z
340 S4=S4+Y*Y*Z
350 S5=S5+B[2]*Y*Z
360 N=N+Z
370 RETURN 0
```

TAPE FILE NO. 5

```
10 DIM C[36],B[8],E$[51]
20 IF N <= D2-W THEN 330
30 D1=6
40 IF W=0 THEN 320
50 T=0
60 FOR I=1 TO D1+1
70 B[I]=0
80 FOR J=1 TO D1-I+2
90 R=(I+J-1)*(D2+2-0.5*(I+J))
100 B[I]=B[I]+C[T+J]*C[R]
110 NEXT J
120 T=I*(D2+(3-I)/2)
130 NEXT I
140 R1=0
150 FOR I=2 TO D1+1
160 R1=R1+C[I*(D2+(3-I)/2)]↑2
170 NEXT I
180 T0=C[(D2+1)*(D2+2)/2]
190 T0=T0-C[D2+1]↑2
200 PRINT
210 PRINT TAB33"COEFFICIENTS"
220 PRINT
```

```
230 FORMAT 30X,F3.0,F12.4
240 FOR I=1 TO D1+1
250 WRITE (15,230)"B("I-1")="B[I]
260 NEXT I
270 PRINT
280 PRINT
290 PRINT TAB28"R SQUARE = "R1/T0
300 PRINT
310 GOTO 730
320 IF N>D2 THEN 350
330 DISP "NOT ENOUGH POINTS"
340 END
350 P=W=1
360 D2=D2+1
370 FOR J=1 TO D2
380 C[P]=SQRC[P]
390 FOR I=1 TO D2-J+1
400 C[P+I]=C[P+I]/C[P]
410 NEXT I
420 R=P+I
430 S=R
440 FOR L=1 TO D2-J
450 P=P+1
460 FOR M=1 TO D2+2-J-L
470 C[R+M-1]=C[R+M-1]-C[P]*C[P+M-1]
480 NEXT M
490 R=R+M-1
500 NEXT L
510 P=S
520 NEXT J
530 T=(D2+1)*(D2+2)/2
540 FOR I=1 TO D2-1
550 T=T-1-I
560 C[T]=1/C[T]
570 FOR J=1 TO D2-I
580 P=D2+1-I-J
590 P=P*(D2+1-(P-1)/2)-I
600 R=P-J
610 S=0
620 U=I+J+1
630 V=P
640 FOR K=1 TO J
650 V=V+U-K
660 S=S-C[R+K]*C[V]
670 NEXT K
680 C[P]=S/C[R]
690 NEXT J
700 NEXT I
710 C[1]=1/C[1]
720 GOTO 50
730 LINK 6
```

TAPE FILE NO. 6

```
10 DIM C[36],B[8],E$[51]
20 A=0
30 B=1
40 C=0.02
50 Y=FNZ0
60 STORE  DATA 8,E
70 LOAD 7
80 DEF FNZ(Z)
90 PRINT
100 I9=0
110 SCALE 0,1,0,1
120 FOR I=A TO B STEP C
130 Y=B[D1+1]
140 FOR J=D1 TO 1 STEP -1
150 Y=Y*I+B[J]
160 NEXT J
170 IF H9=0 THEN 200
180 IF I>(H9/100) THEN 200
190 Y=0
200 PRINT "R="I;TAB20"E(R)="Y
210 I9=I9+1
220 E[I9]=Y
230 PLOT I,Y
240 NEXT I
250 DISP
260 RETURN 0
```

TAPE FILE NO. 7

```
10 DIM E$[51],P$[41]
20 LOAD  DATA 8,E
30 PRINT
40 PEN
50 N=U=0
60 FOR J=1 TO 21
70 P1=R=0
80 FOR I=2 TO 51
90 R=R+0.02
100 X=R*U
110 P=E[I]*R*0.02*FND(X)
120 P1=P1+P
130 NEXT I
140 P1=P1+0.5*P-P
150 P[J]=ABS(P1)
160 IF J=1 THEN 200
170 P[J]=P[J]/P[1]
180 PRINT "U=",U,"F(U)=",P[J]
190 GOTO 210
200 PRINT "U=",U,"F(U)=",1
```

```
210 U=U+0.5
220 NEXT J
230 U2=0
240 P[1]=1
250 SCALE -1,10.5,-0.1,1.05
260 XAXIS 0,1,0,10
270 YAXIS 0,0.1,0,1
280 FOR I=1 TO 21
290 PLOT U2,P[I]
300 U2=U2+0.5
310 NEXT I
320 DISP
330 LABEL (*,1,1,0,7/10)
340 FOR Y=0 TO 1 STEP 0.1
350 PLOT 0,Y,1
360 CPLOT -6,-0.3
370 LABEL (440)Y
380 NEXT Y
390 FOR X=0 TO 10 STEP 1
400 PLOT X,0,1
410 CPLOT -3,-2
420 LABEL (450)X
430 NEXT X
440 FORMAT F4.1
450 FORMAT F4.0
460 END
470 DEF FND(X)
480 Q2=X/2
490 Q1=0
500 Q3=Q4=1
510 FOR Q5=2 TO ABSN
520 Q3=Q3*Q5
530 NEXT Q5
540 Q0=Q3=1/Q3
550 Q3=-Q3*Q2/Q4*Q2/(N+Q4)
560 IF ABSQ3/Q0<1E-10 THEN 610
570 Q0=Q0+Q3*(Q3>0)
580 Q1=Q1+Q3*(Q3<0)
590 Q4=Q4+1
600 GOTO 550
610 E=10+LGTABS(1+Q1/Q0)
620 RETURN (Q0+Q1)*(Q2+(Q2=0)*(N=0))↑N

TAPE FILE NO. 8

DATA STORAGE
```